



Tutorial

Finite Element Analysis with CADINP

FEA Programs Version 21 / 23

Table of contents

| | | |
|----------|--|-----------|
| 1 | Table of contents | 2 |
| 2 | Basic principles of SOFiSTiK | 3 |
| 2.1 | Preface..... | 3 |
| 2.2 | SOFiSTiK modules | 4 |
| 2.3 | Trouble-shooting | 8 |
| 2.4 | Download of the latest program versions | 9 |
| 2.5 | SOFiSTiK in the Internet | 9 |
| 3 | Example frame 2d | 10 |
| 3.1 | Preface..... | 10 |
| 3.2 | Input of materials and cross sections | 11 |
| 3.3 | Input of the system | 11 |
| 3.4 | Input of loads and actions | 12 |
| 3.5 | Input for the computation of the static analysis | 12 |
| 3.6 | Superpositioning with MAXIMA | 13 |
| 3.7 | Input for the design analysis | 14 |
| 3.8 | Compilation of the documentation with URSULA | 15 |
| 3.9 | Data evaluation with WINGRAF | 16 |
| 3.10 | Data evaluation with DBVIEW | 17 |

1 Basic principles of SOFiSTiK

1.1 Preface

To be able to understand the procedural method of SOFiSTiK, the basic program structure is shown in figure 1. It is essential, that any data is written into one central database (SOFiSTiK Database CDB).





Figure 1: FEA Program structure

1.2 SOFiSTiK modules

The analysis of a structural position is done in 3 steps:

- Pre-Processing: Input of structural system, loads, etc.
- Processing: Structural analysis of the system (analysis of stress resultants, support forces, deformations, etc.)
- Post-Processing: Subsequent analyses, such as design of reinforced concrete or steel, as well as the compilation of the alphanumeric and/ or graphic output.

The Pre-Processing can be done either with SOFiPLUS, MONET or TEDDY. SOFiPLUS and MONET are graphical user interfaces. This tutorial focuses on the input with teddy, using the CADINP language.



```

+prog aqua urs:1
head example of concrete frame according to DIN 1045-1
echo mat full
norm din 1045-1 $ Important: Input of the norm for the whole calculation here!
CONC 1 C 30
STEE 2 BST 500SA titl 'reinf. bar - normal ductility'
STEE 3 BST 500MA titl 'reinf. mesh - normal ductility'
STEE 4 BST 500SB TITL 'reinf. bar - high ductility'
srec 1 .6 .4 .2 1.4 mno 1 so 3.5 su 3.5 aso 0 asu 0 mrf 2 $ waling
srec 2 .6 .4 mno 1 so 3.5 su 3.5 aso 0 asu 0 mrf 2 $ Stiele evtl. Plattenbalken z
end

+prog sofimsha urs:2
head System
syst fram $ plane frame in xy-plane
let#L 12.0 $ span [m]
let#H 3.5 $ height [m]
$ input of nodes
node no x y
      1 0 0
      2 0 -#H
      3 #L/2 -#H
      4 #L -#H
      5 #L 0
$ input of node constraints
Node No Fix
  
```

Figure 2: Input with TEDDY using CADINP

The numerical input for any SOFiSTiK module can be done using the text editor TEDDY.

The text editor is directly linked to all help -files with the Adobe Acrobat Reader. They can be opened by pressing F1 and the program will automatically jump to the appropriate input record. Thus you can instantly read any information for the input. Furthermore, the valid input for one record is shown in the TEDDY footer.

The analysis is controlled by the program WinPS. With it you can select and deselect single modules, do a quick-start of the analysis or analyze just a single module. The protocol gives detailed information about the analysis's progress.




Figure 3: WINPS

Ursula is used for the output and documentation of the results. It helps controlling the volume of the output, page numbers, date etc. and also inserting additional text, graphics and pictures.




Figure 4: URSULA

ANIMATOR and WinGRAF are programs to control the system graphically and getting a first view of the results.




Figure 5: ANIMATOR

WinGRAF allows saving graphics of the results. These can be reused for subsequent structural positions or as standards for the bureau.





Figure 6: WINGRAF

DBVIEW is a program that shows data contained in the central database CDB as a table. Here you can export data, e.g. stress resultants to EXCEL.



Printvolume : All results separate
Loadcases : All
Elements : All
Sections : All

Beam Elements

Results of Design

| Elem. | X | LC | Nr | Ni | Myi | Mzi |
|-------|-------|----|------|--------|---------|-------|
| | [m] | | | [kN] | [kNm] | [kNm] |
| 1 | 0,000 | | 1421 | 0,0 | 0,00 | 0,00 |
| | | | 1422 | 0,0 | 0,00 | 0,00 |
| | | | 1425 | 0,0 | 0,00 | 0,00 |
| | | | 1426 | 0,0 | 0,00 | 0,00 |
| | | | 1429 | 0,0 | 0,00 | 0,00 |
| | | | 1430 | 0,0 | 0,00 | 0,00 |
| | | | 1475 | 0,0 | 0,00 | 0,00 |
| | | | 1476 | 0,0 | 0,00 | 0,00 |
| | | | 2121 | -106,3 | -251,08 | 0,00 |
| | | | 2122 | -833,4 | -281,76 | 0,00 |
| | | | 2125 | -144,5 | -315,07 | 0,00 |
| | | | 2126 | -499,1 | -226,16 | 0,00 |
| | | | 2129 | -499,1 | -226,16 | 0,00 |
| | | | 2130 | -144,5 | -315,07 | 0,00 |
| | | | 2175 | -477,4 | -230,63 | 0,00 |
| | | | 2176 | -147,3 | -246,81 | 0,00 |
| | | | 2222 | -833,4 | -281,76 | 0,00 |
| | 0,389 | | 1421 | 0,0 | 0,00 | 0,00 |
| | | | 1422 | 0,0 | 0,00 | 0,00 |
| | | | 1425 | 0,0 | 0,00 | 0,00 |
| | | | 1426 | 0,0 | 0,00 | 0,00 |
| | | | 1429 | 0,0 | 0,00 | 0,00 |
| | | | 1430 | 0,0 | 0,00 | 0,00 |
| | | | 1475 | 0,0 | 0,00 | 0,00 |

Figure 7: BVIEW

1.3 Trouble-shooting

If you have problems, please do the following:

- Check your system with ANIMATOR and WinGRAF . In this way you can easily locate errors in the input.
- We highly recommend checking the plausibility of the results by conducting a rough analysis. Wrong input usually produces wrong results.
- Check, if you use the latest versions of the SOFiSTiK programs.

If there are still have questions or a lack of clarity, please contact the SOFiSTiK hot-line. Please send your request via e-mail to statik@sofistik.de along with the corresponding project file *.dat and, if necessary, also the result file *.erg and tell us the program versions you are using (*.prt).


The SOFiSTiK support team will try to answer your questions as competent and fast as possible at any time and will send you a solution. You can make our work easier by describing your problem as detailed and exactly as possible. Please name one relevant element and one load case, where we can reproduce your problem. It would be best to present your problem by means of a small test example. Thus we can work much more efficiently and faster on your problem and provide you the best solution.

1.4 Download of the latest program versions

Customers with service contracts can download the latest versions of our programs with our freeware program SONAR.

An update with SONAR is done in 4 steps. The current password can be found on your latest service contract invoice. The program guides you through the update process.

1.5 SOFiSTiK on the Internet



2 Example frame 2d




Figure 8: Structural system frame 2d

2.1 Preface

This exercise will guide you through the input of a plane frame and its structural analysis.

We use the SOFiSTiK modules AQUA (materials and cross-sections), SOFiMSHA (geometrical model), SOFiLOAD (input of loads), STAR2 (analysis), MAXIMA (combination rules and superposition), AQB (design) and WinGRAF (graphical output) for the analysis. The documentation will be done with URSULA. In addition you will see different possibilities for the evaluation of the results.

The exercise demonstrates some of the possibilities our programs offer and it should be a first guide for your practice.

The tutorial video about CADINP provides some tips and tricks for using TEDDY. Please also read the SOFiSTiK documentation for detailed information on any module.

2.2 Input of materials and cross sections

In AQUA you have to define the design code for the whole analysis. This input is mandatory. Two different cross sections will be used for the analysis of the frame system shown above.

```
+prog aqua
head example of concrete frame according to DIN 1045-1
echo mat full
norm din 1045-1 $ Important: Input of the norm for the whole
                  $ calculation here!

CONC 1 C 30
STEE 2 BST 500SA titl 'reinf. bar - normal ductility'
STEE 3 BST 500MA titl 'reinf. mesh - normal ductility'
STEE 4 BST 500SB TITL 'reinf. bar - high ductility'
srec 1 .6 .4 .2 1.4 mno 1 so 3.5 su 3.5 aso 0 asu 0 mrf 2
srec 2 .6 .4      mno 1 so 3.5 su 3.5 aso 0 asu 0 mrf 2
end
```

2.3 Input of the system

Nodes and beams will be defined the „classic“ way. With CADINP, SOFiSTiK offers the possibility to define variables and use other programming features.

```
+prog sofimsha
head System
syst fram $ plane frame in xy-plane
let#L 12.0 $ span [m]
let#H 3.5 $ height [m]
$ input of nodes
node    no    x    y
        1     0     0
        2     0    -#H
        3    #L/2    -#H
        4     #L    -#H
        5     #L     0
$ input of node constraints
Node No Fix
      1  f
      5  f
$ input of beams
beam 1 1 2 ncs 2 div 9
      2 2 3 ncs 1 div 8
      3 3 4 ncs 1 div 8
      4 4 5 ncs 2 div 9
end
```

Variables defined with LET are only active within one module, whereas variables defined with #DEFINE are effective in the whole input file.

2.4 Input of loads and actions

There are two options of defining loads. First, you can make the input directly within the analysis modules ASE or STAR. But we highly recommend using SOFILOAD, because the loads can be used for all following calculations with any module. For analyses according to the new standards / codes, it is important to define an action for each load case.

```
+prog sofiload
HEAD Lasten
ACT  G GAMU 1.35 1.0 PSIO 1.0 1.0 1.0 PART G SUP PERM
    LC 1 Type G Titl 'gk1 self weith' DLY 1.0
    LC 2 Type G Titl 'gk2 secondary dead load'
        Beam 2 3 1 Type pyp 0.8*6 $ 6m load collection area
    LC 3 Type G Titl 'e1 earth thrust by onesided fill'
        Beam 1 Type pxp pa 35*6 pe 0 ref yy

ACT  L GAMU 1.50 0.0 PSIO 0.7 0.5 0.3 PART Q SUP COND
        $ psi-values: e.g. category B bureau
    LC 4 Type L Titl 'qk variable load waling'
        Beam 2 3 1 Type pyp 2.0*6 $ e.g. bureaus according to DIN 1055
    LC 5 Type L Titl 'ep earth thrust by traffic'
        Beam 1 type pxp 2.5*6
    LC 6 Type L Titl 'Qk single loads Stiele'
        Node 2 PY 150
            4 Py 150
end
```

2.5 Input for the structural analysis

The analysis of the load cases is done with ASE.

The inclination of the system can easily be defined with OBLI. Please read the SO-FiSTiK documentation for more information.

ATTENTION: Analyses according to 2nd order theory only allow one load case!

```
+prog ase
head 'calculation of single loadcases'
Echo Full no
Echo Lsum yes
syst prob line
LC 1,2,3,4,5,6
end

+prog ase
head serviceability limit state Th. I. order
syst prob line
lc 10 fact 1 dly 1.0 titl 'full load Th. I'
```

```

    lcc 2,3,4,5,6
end

+prog ase
head SLS Th. II. order
syst prob th2 iter 20
OBLI SX=1/200 STOR Sum
lc 11 fact 1 dly 1.0 titl 'full load Th. II'
    lcc 2,3,4,5,6
end

```

2.6 Superpositioning with MAXIMA

Program MAXIMA analyzes the authoritative stress resultants for the design analysis.

```

+prog maxima
head ULS Ultimate limit state
echo Load,Fact,full no
ECHO BEAM yes
echo tabs yes

    COMB 1 EXTR DESI BASE 2100
    ACT  G GAMU 1.35 1.0 PSIO 1.0 1.0 1.0
        LC 1,2,3 G
    ACT  L GAMU 1.50 0.0 PSIO 0.7 0.5 0.3
        LC 4,5,6 Q
    SUPP 1 etyp beam type n  EXTR mami TITL ULS_BEAM
    SUPP 1 etyp beam type vz EXTR mami TITL ULS_BEAM
    SUPP 1 etyp beam type my EXTR mami TITL ULS_BEAM
    SUPP 1 etyp beam type uz EXTR mami TITL ULS_NODE
    SUPP 1 etyp node type py EXTR mami TITL ULS_NODE
    SUPP 1 etyp node type px EXTR mami TITL ULS_

$ Simple input without ACT and LC
$ only the superposition of the normal force is inquired here
$ ECHO LOAD,FACT full. Full output of superposition + safety
$ factors
COMB 2 EXTR DESI BASE 2200 TITL 'simple input'
SUPP 2 EXTR MIN ETYP BEAM TYPE N TITL ULT_BEAM
End

+prog maxima
head SLS service quasi permanent comb.
echo full no ; echo tabs yes
    COMB 4 EXTR PERM BASE 1400 $ input vers. 21
    ACT  G GAMU 1.35 1.0 PSIO 1.0 1.0 1.0
    ACT  L GAMU 1.50 0.0 PSIO 0.7 0.5 0.3
        LC 4,5,6 Q
    SUPP 4 etyp beam type n  EXTR mami TITL SLS_quasi-
permanent_beam $ normal force beam
    SUPP 4 etyp beam type vz EXTR mami TITL SLS_quasi-
permanent_beam $ shear force beam

```

```

SUPP 4 etyp beam type my EXTR mami TITL SLS_quasi -
permanent_beam $ bending moment beam
SUPP 4 etyp beam type uz EXTR mami TITL SLS_quasi -
permanent_node $ displacement
SUPP 4 etyp node type py EXTR mami TITL SLS_quasi -
permanent_node $ support force y
SUPP 4 etyp node type px EXTR mami TITL SLS_quasi -
permanent_node $ support force x
end

```

2.7 Input for the design analysis

For the design of solid structures the ultimate limit state (ULS) and the serviceability limit state (SLS) usually ought to be checked.

```

+prog aqb
head ULS Design without specifics
echo desi yes
$lf 2121,2122,2125,2126,2129,2130 E $ Typ = E for expanded Dar-
stellung mit wing erforderlich e.g. inner lever arm
LC Type (D) $ automatical selection of all loadcases
$ "DESIGN"
desi ulti $ ulti: stress resultant already in ULS
rein rmod save
end

+prog aqb
head SLS robustness min. reinforcement
echo desi yes $ According to DIN 1045-1 (07.01), 13.1.1
(1): min. reinforcement to assure ductile behaviour
echo comb full $ for control of min. crack moment
$lf 1421,1422,1425,1426,1429,1430 Q $ loadcases of SLS
LC TYPE (P) $ quasi permanent loadcases(P)=PERM of SLS
COMB MAMI MY LC1 (P) 0.001 LC2 MCR 1 $ min. crack moment with fctm
desi ulti $ ulti: stress resultants with safety fa-
tors in ULS
rein rmod supe $ supe: superposition with min.
$ reinforcement
end

+prog aqb
head SLS limitation of crack widht, quasi permanent
echo crac yes $ According to DIN 1045-1 (07.01), 11.2.1
$ tab.18
$lc 1421,1422,1425,1426,1429,1430 E $ loadcases of SLS
LC TYPE (P) $ quasi permanent loadcases (P)=PERM of SLS
nstr crac din cw 0.3 chkc 0.45 ksv sl ksb sl $ limitation of
$ crack width rw = 0.3mm
rein rmod supe $ supe: superposition with previous
$ reinforcement
$ The concrete compression stress (quasi permanent) is checked by
$ chkc 0.45 (0.45 fck), although not demanded by DIN 1045-1
end

```

2.8 Compilation of the documentation using URSULA

The output can be compiled with URSULA. It can directly be started out of TEDDY. The tree structure on the left side offers the possibility to switch the results of one module on or off individually.




Figure 9: URSULA with navigation tree and page view

2.9 Data evaluation with WINGRAF




Figure 10: WINGRAF with tree structure and page view

Any data contained in the database CDB can graphically be viewed with WinGRAF. The program offers the possibility to compile individual pages with different graphics, pictures and layers.

One graphic equals one page of printout and can be composed of multiple pictures. Each picture can have multiple layers, thus the nodal and element numbers can be shown in one picture.

These adjustments can be saved in the *.gra file. If you now open WinGRAF out of TEDDY, while the corresponding *.dat is open, this saved *.gra will be opened. Therefore the two files *.dat and *.gra must have the same name.

Another elegant way of using the WinGRAF settings is to include them in the *.dat file as a separate bloc. To do so, please click the right mouse button within the page view of WinGRAF and select „Copy to URSULA“. Then choose what you want to copy.




Figure 11: Copying of the data to TEDDY and URSULA

This data can easily be added to the input file in TEDDY, by clicking the right mouse button and selecting „Paste“. If you now start the analysis over again with some changes, e.g. for the loads, the pictures are automatically compiled within the URSULA file.

2.10 Data evaluation with DBVIEW

DBVIEW is another independent program to read data from the database *.cdb. DBVIEW has to be opened out of TEDDY with the icon in the toolbar on the side. The following dialog box will appear:




Figure 12: Homepage DBVIEW

First you have to select the desired data in the dialog box.

It will be displayed in a list style:




Figure 13: List of selected data

If you want to copy data, e.g. to MS-EXCEL, you can change the output format in Options – Output settings:




Figure 14: Dialog box output format

Afterwards this list can be saved for further use with the command File – Export.